

20. Scale-time

"The dilemma is complete" -Jan Koenderink, [Koenderink1988a]

"It is later than you think" -Chinese proverb

"You are young and life is long and there is time to kill today" -Pink Floyd

```
In[6]:= url = "https://www.romeny.info/FEV-CD/images/";  
SetOptions[ArrayPlot, ColorFunction -> GrayLevel];  
$HistoryLength = 0;  
gD[im_, nx_, ny_,  $\sigma$ ] := DerivativeFilter[im, {ny, nx},  $\sigma$ ];  
Off[DerivativeFilter::arg1];
```

20.1 Introduction

In the time domain we encounter sampled data just as in the spatial domain. E.g. a movie is a series of *frames*, samples taken at regular intervals. In the spatial domain we needed an integration over a spatial area to catch the information. Likewise, we need to have an aperture in time integrating for some time to perform the measurement. This is the *integration time*. Systems with a short resp. long integration time are said to have a fast resp. slow response. Because of the necessity of this integration time, which need to have a finite *duration* (temporal width) in time, a scale-space construct is a physical necessity again.

```
Import[url <> "DaVinciWatch512x540.jpg", ImageSize -> 400]
```



Figure 20.1 This watch Da Vinci Rattrapante (IWC Schaffhausen, Switzerland), named after Leonardo da Vinci who did many inventions in the area of clocks, indicates time at many different time scales, i.e. seconds, minutes, hours, weekdays, day of months, months, years and lunar cycle.

Furthermore, time and space are *incommensurable* dimensions (measurements along these dimensions have different units), so we need a scale-space for space and a scale-space for time.

Time measurements can essentially be processed in two ways: as pre-recorded frames or *instances*, or real-time. Temporal measurements stored for later replay or analysis, on whatever medium, fall in the first category. Humans perform continuously a temporal analysis with their senses, they measure *real-time* and are part of the second category. The scale-space treatment of these two categories will turn out to be essentially different.

Prerecorded sequences can be analyzed in a manner completely analogous with the spatial treatment of scaled operators, we just interchange space with time. The notion of *temporal scale* σ_t then naturally emerges, which is the *temporal resolution*, a device property when we look at the recorded data (it is the inner scale of the data), and a free parameter *temporal scale* when we do the multi-scale analysis.

In the real-time measurement and analysis of temporal data we have a serious problem: the time axis is only a half axis: the past. There is a sharp and unavoidable boundary on the time axis: the present moment. This means that we can no longer apply our standard Gaussian kernels, because they have an (in theory) infinite extent in *both* directions. There is no way to include the future in our kernel, it would be a strong violation of causality.

But there may be a way out when we derive from first principles a new kernel that fulfils the constraint of causality: a kernel defined on a logarithmically remapped time axis. From this new causal kernel we might again derive the temporal and spatio-temporal family of scaled derivative operators. Jan Koenderink [Koenderink1988a] has presented the reasoning to derive the theory, and we will discuss it in detail below.

We will now treat both the pre-recorded and real-time situation in more detail, and give the operational details of the scale-space operators. The best source for reference is Jan Koenderink's original contribution on scale-time [Koenderink1988a].

There have appeared some fine papers discussing the real-time causal scale-space in detail by Luc Florack [Florack1997a, chapter 4.3] and Lindeberg, Fagerström and Bretzner [Lindeberg1996b, Lindeberg1997a, Bretzner1996a, Bretzner1997a]. Lindeberg also discusses the automatic selection of temporal scale [Lindeberg1997b].

20.2 Analysis of prerecorded time-sequences

Prerecorded temporal sample-sequences can be treated just as spatial sample-sequences. The causality constraint is satisfied because we include the whole available axis in our analysis. Boundary effects at both sides of the finite series in a practical situation are dealt with in a proper manner: we *choose* a way to extend the data (recall the discussion in chapter 5). This choice is essentially arbitrary, and the results can be predicted from the choice taken. The most common choice, also the choice taken throughout his book, is the *cyclic* representation: an infinite repetition of the data in all dimensions. Other feasible choices are mirroring, extending with zero's, extending with the mean etc.

The Gaussian derivative kernels with respect to time form the complete family of the *temporal differential operators*. They take the derivatives with respect to time, just as we have seen for the spatial derivative kernels. The temporal scale σ_τ is a free parameter, and the set of results of the operator for a range of scales is called a *temporal scale-space*.

When we combine the class of spatial differential operators with the class of temporal differential operators, we get the complete family of *spatio-temporal operators*. They take simultaneously derivatives to space and time, and can be constructed to any order through the construct of the familiar convolution with the Gaussian kernel.

For each dimension we can define a scale: the temporal scale σ_τ , the spatial scales $\{\sigma_x, \sigma_y, \sigma_z\}$. Typically, the spatial scales are identical (isotropy).

The Gaussian kernels can extend both into the past and the future in this case, because we *know both*.

The full signal is available, the complete recording *duration* is the available time axis, in analogy with the spatial domain of e.g. images. So the kernels extent over both the positive (the 'future') and the negative time axis (the 'past'). There is no real notion of future or past here, such as in the real-time case. The future resp. past here merely refers to data measured later resp. earlier than the datapoint (moment) we are currently analyzing, but which we have already in memory. Here are the graphs of some temporal derivative kernels of low order:

$$\text{gauss} [x_ , \sigma_ / ; \sigma > 0] := \frac{1}{\sigma \sqrt{2\pi}} e^{-\frac{x^2}{2\sigma^2}} ;$$

```
GraphicsRow[Table[Plot[Evaluate[∂tn gauss[t, σ = 1]], {t, -4, 4},
  Ticks → False, PlotStyle → Thick, AxesLabel → {"t", ""},
  PlotLabel → "order: " <> ToString[n]], {n, 0, 3}], ImageSize → 500]
```




Figure 20.2 Proper temporal scale-space kernels for the temporal domain for prerecorded sequences. From left to right: Gaussian temporal derivatives for order 0 to 3. The 0th order kernel is the appropriate sampling kernel without differentiation, the temporal 'point-operator'.

Mixed partial spatio-temporal operators are spatial Gaussian derivative kernels concatenated with temporal Gaussian derivative kernels. This concatenation is a multiplication due to the separability of the dimensions involved.

```
spike = Table[0, {128}, {128}]; spike[[64, 64]] = 108;
gx := ∂D[spike, 1, 0, 20];
lapl := ∂D[spike, 2, 0, 20] + ∂D[spike, 0, 2, 20];
gxt = Table[Evaluate[gx * D[gauss[t, 1.], t]], {t, -4, 4}];
maxgxt = Max[gxt];
laplt = Table[Evaluate[lapl D[gauss[t, 1.], t]], {t, -4, 4}];
maxlaplt = Max[laplt];

p1 = ArrayPlot[#, PlotRange → {-maxgxt, maxgxt}] & /@ gxt;
p2 = ArrayPlot[#, PlotRange → {-maxlaplt, maxlaplt}] & /@ laplt;
GraphicsGrid[{p1, p2}, ImageSize → 400]
Plot[Evaluate[∂t gauss[t, 1.]], {t, -4, 4},
  AxesLabel → {"time →", ""}, AspectRatio → 0.1, ImageSize → 430]
```





Figure 20.3 Spatio-temporal Gaussian derivative kernels. Top row: $\frac{\partial^2 G}{\partial t \partial x}$, the first order derivative in x and t , second row: $\frac{\partial}{\partial t} \left(\frac{\partial^2}{\partial x^2} + \frac{\partial^2}{\partial y^2} \right) G$, the first order time derivative of the spatial Laplacian operator. Bottom plot: First order Gaussian temporal derivative operator, showing the temporal modulation of the spatial kernels. In the second half of the time domain the spatial kernels are reversed in polarity. The horizontal axis is the time axis.

So the appearance of a spatiotemporal operator is as a spatial operator *changing over time*, with a speed indicated ('tuned') by the temporal scale parameter. We

illustrate this with an example in 2D- t . The mathematical expression for the mixed partial derivative operator, first order in the x -direction and time, is $\frac{\partial}{\partial t} \frac{\partial}{\partial x}$. This translates in scale-space theory into a convolution with a concatenation of the Gaussian derivative kernels $\frac{\partial G}{\partial t}$ and $\frac{\partial G}{\partial x}$, leading to a convolution operator $\frac{\partial^2 G}{\partial t \partial x}$ due to the separability of the kernels.

The following commands generate the sequence as an animation (electronic version only). Doubleclick the image to start the animation. Controls appear in the bottom of the notebook window.

```
spike = Table[0, {64}, {64}]; spike[[32, 32]] = 103;
gx := gD[spike, 1, 0, 10];
gxt = Table[Evaluate[gx * D[gauss[t, 1.], t]], {t, -4, 4, .5}];
maxgxt = Max[gxt];
GraphicsRow[frames = ArrayPlot[#, PlotRange → {-maxgxt, maxgxt},
    ImageSize → 100] & /@ gxt, ImageSize → 500]
```



```
ListAnimate[frames]
```




Figure 20.4 Animated sequence of the spatio-temporal Gaussian derivative kernel $\frac{\partial^2 G}{\partial t \partial x}$. In the second half of the time domain the spatial kernels are reversed in polarity. Similar sequence as the top row in figure 20.3.

Note that these temporal derivatives are *point operators*, they merely measure the change of the parameter (e.g. luminance) over time at the operational point, so per pixel. This is essentially different from the detection of motion, the subject of chapter 17, where relations between neighboring pixels are established to measure the *optic flow*.

20.3 Causal time-scale is logarithmic

For real-time systems the situation is completely different. We noted in the introduction that we can only deal with the past, i.e. we only have the half time-axis.

This is incompatible with the infinite extent of the Gaussian kernel to both sides.

With Koenderink's words: "Because the diffusion spreads influences with infinite speed any blurring will immediately spread into the remote future thereby violating the principle of temporal causality. It is clear that the scale-space method can only lead to acceptable results over the complete axis, but never over a mere semi-axis. On the other hand the diffusion equation is the unique solution that respects causality in the resolution domain. Thus there can be no hope of finding an alternative. The dilemma is complete" ([Koenderink1988a]).

The solution, proposed by Koenderink, is to *remap* (reparametrize) the half t -axis into a full axis. The question is then how this should be done. We follow here Koenderink's original reasoning to come to the mapping function, and to derive the Gaussian derivative kernels on the new time axis.

We call the remapping $s(t)$. We define t_0 the present moment, which can never be reached, for as soon as we try to measure it, it is already further in time. It is our reference point, our only point in time absolutely defined, our *fiducial moment*. Every real-time measurement is relative to this point in time. Then s should be a function of $\mu = t_0 - t$, so $s(\mu) = s(t_0 - t)$. We choose the parameter μ to be dimensionless, and $\mu = 0$ for the present moment, and $\mu = -\infty$ for the infinite past. So we get $s(\mu) = s\left(\frac{t_0-t}{\tau}\right)$. The parameter τ is some time constant and is essentially arbitrary. It is the *scale* of our measurement, and we should be able to give it any value, so we want the diffusion to be scale-invariant on the μ -domain.

We also want shift invariance on this time axis, and the application of different clocks, so we require that a transformation $t' = a t + b$ leaves $s(t)$ invariant. μ is invariant if we change clocks.

```
Plot[0, {t, -10, 0}, Ticks -> {{{-6, "μ₂"}, {-2, "μ₁"}}, None},
  PlotRange -> {-1, 5}, AspectRatio -> 0.1,
  Epilog -> {Text["← time", {-10, 0.5}], Text["present=0", {-0.7, 0.9}]},
  ImageSize -> 440]
```



Figure 20.5 The time-axis has only the negative half. The right end is the present moment. The moments in the past μ_1 and μ_2 are observed with a resolution that is proportional with their 'past time', i.e. μ .

On our new time-axis $s(t)$ the diffusion should be a normal, causal diffusion. On every point of the s -axis we have the same amount of diffusion, i.e. the diffusion is *homogeneous* on the s -domain.

The 'inner scale' or resolution of our measurement has to become smaller and

smaller when we want to approach the present moment. But even if we use femtosecond measuring devices, we will never catch the present moment. On the other side of the s -axis, a long time ago, we don't want that high resolution. An event some centuries ago is placed with a resolution of say a year, and the moment that the dinosaurs disappeared from earth, say some 65 million years ago, is referred to with an accuracy of a million years or so.

This intuitive reasoning is an expression of the requirement that we want our time-resolution τ on the s -axis to be proportional to μ , i.e. $\tau \approx \mu$ or $\frac{\tau}{\mu} = \text{constant}$. So for small μ we have a small resolution, for large μ a large one.

```

t0 = 0;
tau = 1;
Plot[-Log[ $\frac{t_0 - t}{\tau}$ ], {t, -2, t0}, PlotRange -> {-1.5, 4},
  AxesLabel -> {"t", "s"}, Epilog -> {Dashing[{0.01, 0.01}],
    Line[{{1, -1.5}, {1, 4}}]}, ImageSize -> 300, PlotTheme -> "Detailed"]
```




Figure 20.6 The logarithmic mapping of the horizontal t -time half-axis onto the vertical s -time full axis. The present moment t_0 (at $t = 1$ in this example, indicated by the vertical dashed line) can never be reached. The s -axis is now a full axis, and fully available for diffusion. The rectangular inset box with gridlines shows a typical working area for a real-time system. The response time delimits the area at the right, the lifetime (history) at the left. Figure adapted from [Florack1997a].

On the s -axis we should have the possibility for normal, causal diffusion. This means that the 'magnification' $\left| \frac{ds}{d\mu} \right|$ should be proportional to $\frac{\tau}{\mu}$. Then the s -axis is 'stretched' for every μ in such a way that the scale (or 'diffusion length' as Koenderink calls it) in the s -domain is a *constant relative* diffusion length in the μ -domain.

Uniform sampling in the s -domain gives a *graded resolution history* in the t - or μ -domain. In formula: $\left| \frac{ds}{d\mu} \right| \approx \frac{\tau}{\mu}$ or $\left| \frac{ds}{d\mu} \right| = \frac{\alpha}{\mu}$. From this partial differential equation we derive that the mapping $s(\mu)$ must be logarithmic:

$$\text{DSolve} \left[\partial_{\mu} s[\mu] == \frac{\alpha}{\mu}, s[\mu], \mu \right]$$

$$\{ \{ s[\mu] \rightarrow c_1 + \alpha \text{Log}[\mu] \} \}$$

So our mapping for s is now: $s = \alpha \ln\left(\frac{t_0-t}{\tau}\right) + \text{constant}$. The constant is an arbitrary translation, for which we defined to be invariant, so we choose this constant to be zero. We choose the arbitrary scaling parameter α to be unity, so we get

$$s = \ln\left(\frac{t_0-t}{\tau}\right).$$

This is a fundamental result. For a causal interpretation of the time axis we need to sample time in a logarithmic fashion. It means that the present moment is mapped to infinity, which conforms to our notion that we can never reach it. We can now freely diffuse on the s -axis, as we have a well defined scale at all moments on our transformed time axis. This mapping also conforms to our notion of resolution of events in the past: our memory seems to do the same weighting of the aperture over the past as the transformation we introduced above.

In the s -domain we can now run the diffusion equation without violation of temporal causality. The diffusion equation is now $\frac{\partial^2 L}{\partial s^2} = \frac{\partial L}{\partial v_s}$, where $v_s = \frac{1}{2} \tau^2$ is the temporal variance. We recall from the diffusion equation in the spatial domain that the Laplacian (of the luminance in our case) along the diffusion axis is equal to the rate of change (of the luminance) with the *variance* of the scale. The check of dimensions on both sides of the equation is always a good help. The temporal blurring kernels (the scale-space measurement apertures) are now given by

$$K(s, s'; \sigma_s) = \frac{1}{\sqrt{2\pi\tau^2}} e^{-\frac{(s-s')^2}{2\tau^2}}, \text{ or } K(s, s'; v_s) = \frac{1}{\sqrt{4\pi v_s}} e^{-\frac{(s-s')^2}{4v_s}}.$$

20.4 Other derivations of logarithmic scale-time

Florack [Florack1997a] came to the same result from a different perspective, from abstract mathematics. He used a method from *group theory*. Essentially, a group is a mathematical set of operations, which (in popular words) all do the same operation, but with a different parameter. Examples of a transformation group are the group of rotations, the group of additions, the group of translations etc. A group is formally defined as a set of similar transformations, with a member that does the unity operation (projects on itself, i.e. does nothing, e.g. rotation over zero degrees, an enlargement of 1, a translation of zero etc.), it must have an inverse (e.g. rotation clockwise, but also anti-clockwise) and one must be able to concatenate its members (e.g. a total rotation which consists of two separate rotations after each other).

Florack studied the group properties of whole and half axes of real numbers. The group of summations is a group on the whole axis, which includes the positive and

the negative numbers. This group however is *not* a group on the half axis. For we might be able to do a summation which has a result outside the allowed domain. The group of multiplications however is a group on the positive half axis.

Two numbers multiplied from the half axis give a result on the same half axis. If we could make all sums into multiplications, we would have an operation that makes it a group again. The formal transformation from sums into multiplications is the logarithmic function: $e^{a+b} = e^a * e^b$ and its inverse $\ln(a * b) = \ln(a) + \ln(b)$. The zero element is addition of zero, or multiplication with one. So the result is the same logarithmic function as the function of choice for the causal parametrization of the half axis.

Lindeberg and Fagerström [Lindeberg1996b] derived the causal temporal differential operator from the non-creation of local extrema (zero-crossings) with increasing scale. Jaynes' method of 'transformation groups' to construct prior probabilities has a strong similarity to the reasoning in this chapter. For details see [Jaynes1968].

Interestingly, we encounter more often a logarithmic parametrization of a half axis when the physics of observations is involved:

- Light intensities are only defined for positive values, and form a half axis. It is well known e.g that the eye performs a logarithmic transformation on the intensity measured on the retina.
- Sound intensities are measured in decibels (dB), i.e. on a logarithmic scale.
- Scale is only defined for positive values, and form a half axis (scale-space). The natural scalestep τ on the scale-axis in scale-space is the logarithm of the diffusion scale σ : $\tau = \ln(\sigma) - \ln(\sigma_0)$ (recall chapter 1).

Another example of the causal logarithmic mapping of the time axis is the striking Law of Benford, which says that in a physical *measurement* that involves a duration, the occurrence of the first digit has a logarithmic distribution. I.e. a "1" as the first digit occurs roughly 6.5 times more often than a "9"! This law follows immediately from the assumption that random intervals are *uniformly* distributed on the logarithmic s -axis [Florack1997a, pp. 112].

Task 20.0 The decay times of a large set of radioactive isotopes is available on internet: isotopes.lbl.gov and ie.lbl.gov/tori.html. Show that the first digits of these decay times indeed show Benford's Law, with a much pronounced occurrence of "1"s. [Buck1993].

Our perception of time also seems to be non-linear. We have more trouble remembering accurate events that took place a long time ago, as in our childhood, than events that just recently happened. Why has last year passed by so quickly, and did a year at highschool seem to last so much longer? Our perception is that time seems to go quicker when we get older. We have a increasingly longer reference to

the past with which we can compare, and our notion is that time seemed to go slower a longer time ago. These observations suggest some logarithmic scaling: stretched out at the low end and compressed at the high end. It is known in psychological literature that our perceived time 'units' may be related to our age: a 10% of age for a 5-year old is half a year, for a 50-year old it is 5 years. A half year seems to pass just as quick for the 5-year old as 5 years for the 50-year old. This also leads to a logarithmic notion of time. See also [Gibbon18981] and [Pöppel1978a].

20.5 Real-time receptive fields

We have now all information to study the shape of the causal temporal derivative operators. The kernel in the transformed s -domain was given above. The kernel in the original temporal domain t becomes

$$K(t, t_0; \tau) = \frac{1}{\sqrt{2\pi} \tau} e^{-\frac{1}{2\tau^2} \ln\left(\frac{t_0-t}{\tau}\right)^2}.$$

In figure 20.7 we see that the Gaussian kernel and its temporal derivatives are *skewed*, due to the logarithmic time axis remapping. It is clear that the present moment t_0 can never be reached.

`Clear [gt];`

`GraphicsRow [Table [τ = 0.2;`

`t0 = 0;`

`Plot [Evaluate [gt [n] = ∂{t,n} $\frac{e^{-\frac{\text{Log}[\frac{t_0-t}{\tau}]^2}{2\tau^2}}}{\sqrt{2\pi}\tau}$], {t, -0.4, 0},`

`PlotRange → All, PlotLabel → "temporal order = " <> ToString[n],
Epilog → Text ["time →", {-0.45, 0}], {n, 0, 2}], ImageSize → 500]`




Figure 20.7 Left, middle, right: the zeroth, first and second Gaussian temporal derivative operator in causal time. The timescale in each plot runs from the past on the left, to the present on the right. The temporal scale $\tau = 200$ ms, the right boundary is the present, $t_0 = 0$. Note the pronounced skewness of the kernels.

The zerocrossing of the first order derivative (and thus the peak of the zeroth order kernel) is just at $t = -\tau$. The extrema of the first derivative kernel are found when we set the second derivative of the time kernel to zero. Here are both solutions:

```
Clear[t, τ, timekernel]; t_0 = 0;
timekernel[t_] =  $\frac{1}{\sqrt{2\pi}\tau^2} \text{Exp}\left[-\frac{1}{2\tau^2} \text{Log}\left[\frac{t_0 - t}{\tau}\right]^2\right]$ ;
zerot = Solve[∂ttimekernel[t] == 0, t]
{{t → -τ}}
```

The extrema of the first derivative kernel are found when we set the second derivative of the time kernel to zero.

```
Simplify[∂t,ttimekernel[t] == 0]

$$\frac{e^{-\frac{\text{Log}\left[-\frac{t}{\tau}\right]^2}{2\tau^2}} \left(-\tau^2 + \tau^2 \text{Log}\left[-\frac{t}{\tau}\right] + \text{Log}\left[-\frac{t}{\tau}\right]^2\right)}{t\tau} = 0$$

```

For these extrema of the second order derivative we need to assist *Mathematica* a little. We replace $\text{Log}\left[-\frac{t}{\tau}\right] \rightarrow \xi$, and reverse the replacement at the end. We now get the extrema for the second order time-derivative:

```
soltt = Solve[∂t,ttimekernel[t] == 0 /. Log[-t/τ] → ξ, ξ]
{{ξ →  $\frac{1}{2}(-\tau^2 - \tau\sqrt{4+\tau^2})$ }, {ξ →  $\frac{1}{2}(-\tau^2 + \tau\sqrt{4+\tau^2})$ }}
zerott = Solve[Log[-t/τ] == ξ, t, Reals] /. soltt
{{{t →  $-e^{\frac{1}{2}(-\tau^2 - \tau\sqrt{4+\tau^2})} \tau$ }}, {{t →  $-e^{\frac{1}{2}(-\tau^2 + \tau\sqrt{4+\tau^2})} \tau$ }}}
```

For $\tau = 0.2$ as in the figure above:

```
zerott /. τ → 0.2
{{{t → -0.160344}}, {t → -0.239682}}
```

It is now easy to combine the spatial and temporal kernels. Time and space are separable incommensurable dimensions, so we may apply the operators in any order: first a spatial kernel and then a temporal kernel is the same as first a temporal and then a spatial kernel. The most logical choice is the simultaneous action. When we make a physical realization of such an operator, we have to plot it in the spatial and time domain. An example is given below for a 2D first order spatial derivative, and first order temporal derivative. We get a 2D-time sequence:

```

spike = Table[0, {128}, {128}]; spike[[64, 64]] = 108;
τ = 0.3-; gt = ∂ttimekernel[-t]; gx = gD[spike, 1, 0, 20];
max = Max[gx ∂ttimekernel[t] /. First[zerott]];
p1 = Table[ArrayPlot[gt gx, PlotRange → {-max, max},
  DisplayFunction → Identity], {t, 0.1-, 0.6-, 0.05-]];
GraphicsRow[p1, ImageSize → 500]

```





Figure 20.8 Eleven frames of a time series of a 2D-time spatio-temporal time-causal Gaussian derivative kernel, which takes the first order derivative to time and the first order to space in the x -dimension. The present moment is on the left. Note the inversion that takes place in the spatial pattern during the time course.

The time series of a 2D-time spatio-temporal time-causal Gaussian derivative kernel is asymmetrical.

```
ListAnimate[p1]
```



This has been measured in cortical simple cells by single cell recordings (see figure 20.9). See for the methodology to map receptive field sensitivity profiles chap. 9 and figure 11.12.

```
In[17]:= GraphicsGrid[Partition[set =
      Import /@ (url <> "separablext" <> ToString[#] <> ".gif" & /@ Range[60]),
      10], ImageSize -> 500]
```

Out[17]=





Figure 20.10 Time sequence of a cortical simple cell. Frames row-wise, upper left frame is first. Time domain: 0 - 300 msec in 5 msec steps. Data of the cell: on-center X cell, non-lagged; X-Y domain size: 3 x 3 degs; Bar size: 0.5 x 0.5 degs; Eye: left; Total time for receptive field measurement: 19 min. Number of repetitions: 50; Duration of stimuli: 26.3ms. From: neurovision.berkeley.edu/Demonstrations/VSOC/teaching/RF/LGN.html.

```
setR = ImageResize[#, 150] & /@ set;
ListAnimate[setR]
```

Out[19]=



20.6 A scale-space model for time-causal receptive fields

The temporal behaviour of cortical receptive fields has been measured extensively. The method most often used is the method of 'reverse correlation' (explained in section 9.6).

Figure 20.10 shows a recording of a cortical simple cell from Freemans' lab in Berkeley, that shows clearly the spatial similarity to the first Gaussian derivative kernel, and the modulated amplitude over time. The spatio-temporal relations become more clear when they are plotted in the spatio-temporal domain. Figure 20.11 shows a plot where the horizontal axis displays the spatial modulation, the vertical axis displays the temporal modulation.

```
In[20]:= Show[Import[url <> "SimpleCell-V1-XT.gif"], Frame → True,
  FrameLabel → {"space →", "time →"}, FrameTicks → None, ImageSize → 200]
Out[20]=
```




Figure 20.11 Spatio-temporal response of a cortical simple cell. From [DeAngelis1995a]. The polarity of the spatial response reverses sign during the time course of the response of the cell. Horizontal axis is space (6 degs), vertical axis is time (0-300 msec). Time zero is at the bottom. Cell data: X-Y domain size: 6 x 6 degs; X-Y grid: 20 x 20 points; time domain: 0 - 300 msec in 5 msec steps; orientation: 15 degs; bar size: 2.5 x 0.5 degs; duration of individual stimuli: 52.8 msec (4 frames). From [neurovision.berkeley.edu]. Copyright Izumi Ohzawa, UC Berkeley.

Here is another cell measured by Dario Ringach and coworkers [Ringach1997].
Data from manuelita.psych.ucla.edu/~dario/research.htm:

```
In[22]:= GraphicsGrid[Partition[
  Import /@ (url <> "Ringach-V1-" <> ToString[#] <> ".gif" & /@ Range[12] ),
  6], ImageSize -> 500]
```

Out[22]=




Figure 20.12 Timescale 35 ms (first frame) to 90 ms (last frame) in 5 ms intervals. Upper left frame is the start of the sequence.

Recent more precise measurements of the spatio-temporal properties of macaque monkey and cat LGN and cortical receptive fields give support for the scale-time theory for causal time sampling. De Valois, Cottaris, Mahon, Elfar and Wilson [DeValois2000] applied the method of reverse correlation and multiple receptive field mapping stimuli (m - sequence, maximum length white noise stimuli) to map numerous receptive fields with high spatial and temporal resolution. Fig 20.13 shows some resulting receptive field maps:

```
In[25]:= GraphicsRow[Import /@
  (url <> "ValoisRF-" <> ToString[#] <> ".jpg" & /@ {a, b, c, d, e, f}),
  ImageSize -> 500]
```

Out[25]=




Figure 20.13 Examples of spatio-temporal receptive field maps of a sample of V1 simple cells of macaque monkey. Vertical axis in each plot: time axis from 0 ms (bottom) to 200 ms (top). Horizontal axis per plot: space (in degrees), a: 0-0.9, b: 0-1.2, c,d: 0-0.6, e: 0,1.9, f: 0-1.6 degrees. Note the clearly skewed sensitivity profiles in the time direction. Every 'island' has opposite polarity to its neighboring 'island' in each plot. Due to black and white reproduction the sign of the response could not be reproduced. The scale-space models for the plots are respectively: a: $\frac{\partial^2 L}{\partial x^2}$, b: $\frac{\partial^3 L}{\partial x^3}$, c: $\frac{\partial L}{\partial x}$, d: $\frac{\partial L}{\partial x}$, e: $\frac{\partial^3 L}{\partial x^2 \partial t}$, f: $\frac{\partial^2 L}{\partial x \partial t}$. Adapted from [DeValois2000].

If we plot the predicted sensitivity profiles according to Gaussian scale-space theory we get remarkably similar results. In figure 20.14 the space-time plots are shown for zeroth to second spatial and temporal differential order. Note the

skewness in the temporal direction.

Important support for especially the Gaussian scale-time derivative model comes from another observation by De Valois et al. [DeValois2000]. They state: 'Note that the response time course of these two non-directional cell populations are approximately 90 degrees shifted in phase relative to each other, that is, they are on average close to temporal quadrature.

That is, the population of biphasic cells is shifting over from one phase to the reverse at the time that the monophasic cell population reaches its peak response' (a quadrature filter can be defined, independently of the dimensionality of the signal space, as a filter that is zero over one half of the Fourier space. In the spatial domain, the filter is complex: an even real part and an odd imaginary part).

```
In[30]:= Clear [gt, gs, n];  $\tau = 0.3$ ;  $t_0 = 0$ ;  $\sigma = 2$ ;  
gt [n_] = D [  $\frac{1}{\sqrt{2\pi}\tau^2} \text{Exp} \left[ -\frac{1}{2\tau^2} \text{Log} \left[ -\frac{t_0 - t}{\tau} \right]^2 \right]$ , {t, n} ];
```

```
In[38]:= gs[n_] = D[ $\frac{1}{\sqrt{2\pi}\sigma^2} \text{Exp}\left[-\frac{1}{2\sigma^2} x^2\right]$ , {x, n}];
```

```
GraphicsRow[Table[ContourPlot[Evaluate[gt[i] × gs[j]],
  {x, -15, 15}, {t, .01, .8}, PlotPoints → 30,
  ColorFunction → "Pastel", FrameLabel → {"space", "time"},
  PlotLabel → "nspace=" <> ToString[j] <> ", ntime=" <> ToString[i] ],
  {j, 0, 2}, {i, 0, 2}], ImageSize → 500]
```

Out[39]=




Figure 20.14 Model for time-causal spatio-temporal receptive field sensitivity profiles from Gaussian scale-space theory. All combinations from zeroth to second order partial derivative operators with respect to space and time are shown. Vertical axis: time. Horizontal axis: space.

This fits very well to the model: The receptive fields are the zeroth and first order temporal Gaussian derivative, and for these functions the zero crossing of the the first order derivative coincides with the maximum of the zeroth order.

```
In[41]:= GraphicsRow[Import /@ {url <> "ValoisTimeToPeak01.jpg",
  url <> "ValoisTimeToPeak02.jpg"}, ImageSize -> 500]
```

```
Out[41]=
```




Figure 20.15 Distribution of time-to-peak responses, calculated from the spatio-temporal receptive field mappings of the population of monophasic cells, and of both the initial and later reversed phase of the biphasic cells. Note that the time of peak response for the monophasic cells almost exactly coincides with the time of polarity reversal of the biphasic cells, in perfect agreement with the Gaussian temporal derivative model. Horizontal axis: time-to-peak response time, in ms. Vertical axis: number of cells. Right figure: Time-to-peak response of a random sample of just 5 cells, showing that even a small sample set has the properties of the larger collection. Adapted from [DeValois2000].

Task 20.0 Give a measure for the skewness of the time-causal Gaussian kernel.

20.7 Conclusion

The causal time-scale, multi-scale temporal differential operator model from Gaussian scale-space theory has not yet been tested against the wealth of currently available receptive field measurement data.

It may be an interesting experiment, to test the quantitative similarity, and to find the statistics of the applied spatial and temporal scales, as well as the distribution of the differential order.

The Gaussian scale-space model is especially attractive because of its robust physical underpinning by the principal of temporal causality, leading to the natural notion of the logarithmic mapping of the time axis in a real-time measurement.

The distributions of the locations of the different scales and the differential orders have not yet been mapped on the detailed cortical orientation column with the pinwheel structure. Orientation has been clearly charted due to spectacular developments in optical dye high resolution recording techniques in awake animals. Many interesting questions come up: Is the scale of the operator mapped along the spokes of the pinwheel? Is the central singularity in the repetitive pin-

wheel structure the largest scale? Is differential order coded in depth in the columns?

These are all new questions arising from a new model. The answer to these questions can be expected within a reasonable time, given the fast developments, both in high resolution recording techniques, and the increase in resolution of non-invasive mapping techniques as high-field functional magnetic resonance imaging (fMRI) [Logothetis1999].

20.8 Summary of this chapter

When a time sequence of data is available in stored form, we can apply the regular symmetric Gaussian derivative kernels as causal multi-scale differential operators for temporal analysis, in complete analogy with the spatial case. When the measurement and analysis is real-time, we need a reparametrization of the time axis in a logarithmic fashion. The resulting kernels are skewed towards the past. The present can be never reached, the new logarithmic axis guarantees full causality. The derivation is performed by the first principle of a scale of observation on the new time axis which is proportional to the time the event happened. This seems to fit well in the intuitive perception of time by humans.

Recent physiological measurements of LGN cell receptive fields and cortical V1 simple cell receptive fields reveal that the biological system seems to employ the temporal and spatiotemporal differential operators. Especially striking is the skewness in the temporal domain, giving strong support for the working of the biological cells as time-causal temporal differential operators.